

Date e ore *differenza e comparazione*



Marco Riva



www.markonetools.it

IBM **CHAMPION**

2021

Power Systems area



Ultimo aggiornamento: 27/11/2021

Abbiamo parlato di...

- ▀ definizione campi e variabili
- ▀ definizione del formato
- ▀ uso delle costanti data/ora
- ▀ reperimento data/ora correnti
- ▀ conversione tra tipi dati
- ▀ conversione di durate
- ▀ somma e sottrazione di durate

Power
coffee
29/2021

Power
coffee
32/2021



2

Oggi parliamo di...

- comparazione
- differenza



3

Riepilogo definizioni date/ore

	Description	RPG IV			DDS		SQL		Value	Format
		Type ¹	Length	Keyword	Type	Keyword	Type	Length		
DATE/TIME ²	Date	D <i>date</i>	6 8 10	datfmt(*JUL) ³ datfmt(*YMD) datfmt(*DMY) datfmt(*MDY) datfmt(*ISO) ⁴ datfmt(*EUR) datfmt(*USA) datfmt(*JIS)	L	DATFMT DATSEP	DATE ⁵		40/001 to 39/365, def. 40/001 40/01/01 to 39/12/31 def. 40/01/01 01/01/40 to 31/12/39 def. 01/01/40 01/01/40 to 12/31/39 def. 01/01/40 01/01/0001 to 31/12/9999 def. 01/01/0001 01.01.0001 to 31.12.9999 def. 01.01.0001 01/01/0001 to 12/31/9999 def. 01/01/0001 0001-01-01 to 9999-12-31 def. 0001-01-01	2 digits yy/ddd 2 digits yy/mm/dd 2 digits dd/mm/yy 2 digits mm/dd/yy 4 digits yyyy-mm-dd 4 digits dd.mm.yyyy 4 digits mm/dd/yyyy 4 digits yyyy-mm-dd
	Time	T <i>time</i>	8	timfmt(*HMS) ⁶ timfmt(*ISO) timfmt(*USA) timfmt(*EUR) timfmt(*JIS)	T	TIMFMT TIMSEP	TIME ⁷		00:00:00 to 24:00:00, def. 00:00:00 00.00.00 to 24.00.00, def. 00:00:00 00:00 AM to 12:00 AM, def. 00:00 AM 00.00.00 to 24.00.00, def. 00:00:00 00.00.00 to 24.00.00, def. 00:00:00	hh:mm:ss hh.mm.ss hh:mm AM or hh:mm PM hh.mm.ss hh:mm:ss
	Timestamp	Z <i>timestamp</i>	26-32 ⁸	*ISO	Z		TIMESTAMP ⁹		0001-01-01-00.00.00.000000 to 9999-12-31-24.00.00.000000 def. 0001-01-01-00.00.00.000000	YYYY-MM-DD- hh.mm.ss.mmmmmm ¹⁰

- ¹ In *brown italic font* the free-form syntax
- ² A 0 after format Indicates NO separator (ex. *ISO0)
- ³ Valid separator for 2 digits format: / - . , &
- ⁴ The default format for date, time e timestamp fields is *ISO
- ⁵ The internal representation is a string of 4 bytes that contains an integer (scaliger number)
- ⁶ Valid separator: . , &
- ⁷ The internal representation is a string of 3 bytes that contains two packed decimal digits.
- ⁸ From 7.2 the lenght can be until 32
- ⁹ The internal representation is a string of 3 of between 7 and 13 bytes
- ¹⁰ The fractional part is facultative. The length could be from 1 to 12. The default is 6.

IBM i 7.3



Poster tipi dati:

<https://www.markonetools.it/ce-ne-di-tutti-i-tipi/>

Comparazione di due campi data/ora

- Per comparare due variabili di tipo data/ora **NON** è strettamente necessario che siano nello stesso formato
- Il formato è una rappresentazione del valore data, quindi anche cambiando formato il valore contenuto nel campo non cambia

```
dcl-s ISO_date date(*ISO);  
dcl-s MDY_date date(*MDY);  
  
ISO_date = d'2015-10-22';  
MDY_date = d'2015-10-22';  
if (ISO_date = MDY_date);  
  // eseguo queste operazioni  
else;  
  // no!  
endif;
```

la costante data va
scritta sempre nel
formato di default
del programma



Differenza con RPG – old way

- Nel vecchio RPG in formato fisso si utilizza il codice operativo SUBDUR.
- *Fattore 1* e *2* devono essere due campi data, ora o timestamp
- Il campo *risultato* deve essere un campo numerico senza decimali e dimensione massima 15 digit e deve essere seguito da : e un codice di durata (*YEARS o *Y, *MONTHS o *M, *DAYS o *D, *HOURS o *H, *MINUTES o *MN, *SECONDS o *S, *MSECONDS o *MS)

```
dcl-s DataOrdine date inz(d'2019-10-01');  
dcl-s DataConsegna date inz(d'2019-10-20');  
dcl-s DiffGG packed(3);
```

```
c      DataConsegna  subdur      DataOrdine      DiffGG:*D
```

Differenza con RPG – new way

- In RPG formato libero si usa la funzione %diff (introdotta da V5R1)

```
%diff(op1 : op2 : unit { : frac })
```

 un codice di durata (*days, *hour...)

- il *risultato* è sempre **arrotondato in difetto** e l'eventuale **resto** viene **ignorato**.

P.es. una differenza di 61 minuti è uguale a 1 ora, invece una differenza di 59 minuti è uguale a 0 ore.

```
dcl-s DataOrdine date inz(d'2019-10-01');  
dcl-s DataConsegna date inz(d'2019-10-20');  
dcl-s DiffGG packed(3);
```

```
DiffGG = %diff(DataConsegna : DataOrdine : *days);
```

Differenza con SQL/1

- ▶ attenzione al calcolo della differenza con le funzioni scalari year, month o day.
Si potrebbe essere tentati di pensare di utilizzare la funzione SQL che corrisponde al terzo parametro della funzione RPG %diff, ovvero month se si desidera la differenza in mesi
- ▶ ➔ **NO!** Le funzioni year, month o day estraggono la porzione di anno, mese o giorno da una data.



Differenza con SQL/2

- Per calcolare la differenza in giorni tra due date bisogna utilizzare la funzione days che restituisce il numero di giorni che intercorrono tra l'argomento e la data del 1 gennaio 0001

```
dcl-s DataOrdine date inz(d'2018-11-01');  
dcl-s DataConsegna date inz(d'2019-10-20');  
dcl-s DiffGG packed(3);
```

```
exec sql  
    set :DiffGG = days(:DataConsegna) - days(:DataOrdine);  
// equivalente a:  
DiffGG = %diff(DataConsegna : DataOrdine : *days);
```

```
exec sql  
    set :DiffMM = month(:DataConsegna) - month(:DataOrdine);  
// !!! non equivale a:  
DiffMM = %diff(DataConsegna : DataOrdine : *months);
```



Differenza con SQL/3

- Una funzione SQL simile a `days` è `julian_day` che restituisce il numero di giorni a partire dal 1 gennaio 4713 a.c. (ovvero l'inizio del calendario giuliano)
- Quindi la differenza tra due date convertite con la funzione `julian_day` restituisce il numero di giorni intercorrenti tra le due date.

```
dcl-s DataOrdine date inz(d'2018-11-01');  
dcl-s DataConsegna date inz(d'2019-10-20');  
dcl-s DiffGG packed(3);
```

```
exec sql  
  set :DiffGG = julian_days(:DataConsegna) - julian_days(:DataOrdine);
```




10

Differenza "composta" con SQL

- ▶ facendo con SQL una sottrazione tra due date si ottiene un risultato "composto" da un numero che rappresenta la differenza in 3 porzioni: aammgg
- ▶ il risultato ottenuto va analizzato **a partire da destra:** 
 - ▶ 2 byte per il numero di giorni
 - ▶ 2 byte per il numero di mesi
 - ▶ la restante porzione è il numero di anni

```
dcl-s DataOrdine date inz(d'2009-12-01');  
dcl-s DataConsegna date inz(d'2019-01-20');  
dcl-s Diff char(10);  
  
exec sql  
    set :Diff = :DataConsegna - :DataOrdine;  
// risultato = 90119 ovvero 9 aa 01 mm 19 gg
```



Differenza in mesi con SQL/1

- La funzione SQL months between restituisce un numero stimato di mesi tra due date
- Il *risultato* è un tipo dati dec (31, 15) .
- Si assume che il mese abbia 31 giorni, ma la funzione tiene anche conto se si sta calcolando la differenza tra gli ultimi giorni di mesi differenti



Differenza in mesi con SQL/2

la porzione giorno delle due date è identica (29) e quindi il numero stimato di mesi differenza è 1

per entrambe le date si tratta dell'**ultimo giorno del mese** e quindi il numero stimato di mesi di differenza è 1

```
dcl-s DataOrdine date inz(d'2018-11-01');
dcl-s DataConsegna date inz(d'2019-10-20');
dcl-s DataOrdine2 date inz(d'2009-12-01');
dcl-s DataConsegna2 date inz(d'2019-01-20');
dcl-s DataOrdine3 date inz(d'2008-02-29');
dcl-s DataConsegna3 date inz(d'2008-03-29');
dcl-s DataOrdine4 date inz(d'2008-02-29');
dcl-s DataConsegna4 date inz(d'2008-03-31');
dcl-s DiffMM2 packed(31:15);
```

```
exec sql
  set :DiffMM2 = months_between(:DataConsegna, :DataOrdine);
// risultato = 11.61290322580645
```

```
exec sql
  set :DiffMM2 = months_between(:DataConsegna2, :DataOrdine2);
// risultato = 109.6129032258064
```

```
exec sql
  set :DiffMM2 = months_between(:DataConsegna3, :DataOrdine3);
// risultato = 1.0000000000000000
```

```
exec sql
  set :DiffMM2 = months_between(:DataConsegna4, :DataOrdine4);
// risultato = 1.0000000000000000
```



13

Differenza con timestampdiff (SQL)/1

- La funzione SQL timestampdiff restituisce un numero stimato di intervalli del tipo definito dal primo argomento, basato sulla differenza tra due timestamp

`timestampdiff(interval-type, string-expression)`

- interval-type*: specifica il tipo di intervallo da restituire come differenza tra due timestamp (secondi, minuti, ore, giorni...)
- string-expression*: è la **differenza** tra due timestamp convertita in una stringa di 22 caratteri
- risultato*: è un numero intero
Il mese è considerato di 30 giorni e l'anno di 365 giorni o 52 settimane

interval-type	descrizione
1	microsecondi
2	secondi
4	minuti
8	ore
16	giorni
32	settimane
64	mesi
128	trimestri
256	anni



Differenza con timestampdiff (SQL)/2

```
dcl-s DataOrdine date inz(d'2018-11-01');  
dcl-s DataConsegna date inz(d'2019-10-20');  
dcl-s DiffInt int(20);  
  
// differenza in mesi. Risultato: 11 mesi  
exec sql  
  set :DiffInt = timestampdiff(64,  
    cast(cast(:DataConsegna as timestamp) -  
    cast(:DataOrdine as timestamp) as char(22)));  
  
// differenza in giorni. Risultato: 349 giorni  
exec sql  
  set :DiffInt = timestampdiff(16,  
    cast(cast(:DataConsegna as timestamp) -  
    cast(:DataOrdine as timestamp) as char(22)));
```



Riferimenti



➤ E-mail aziendale: mriva@sirio-is.it



➤ Blog: www.markonetools.it



➤ E-mail blog: info@markonetools.it



➤ LinkedIn: www.linkedin.com/in/marcoriva-mk1



➤ Twitter: [@MarcoRiva73](https://twitter.com/MarcoRiva73)



➤ Facebook: <https://www.facebook.com/markonetools/>



➤ YouTube: <https://www.youtube.com/channel/UCb47YJQJCzU-5x4nnGzDu-w>

Power coffee - MK1



16